

18.4 ToolPanel (gb.form)

Das Steuerelement ToolPanel implementiert eine Toolbox mit mehreren vertikal scrollbaren Toolbars. Die Klasse ist erzeugbar. So erzeugen Sie ein neues ToolPanel:

```
Dim hToolPanel As ToolPanel
hToolPanel = New ToolPanel ( Parent As Container ) [ As "EventName" ]
```

Die Klasse verhält sich wie ein Nur-Lesen-Array und gibt ein Objekt zurück, das eine der Toolboxes repräsentiert.

```
Dim hToolPanel As ToolPanel
Dim hToolPanelContainer As _ToolPanelContainer
hToolPanelContainer = hToolPanel [ Index As Integer ]
```

18.4.1 Eigenschaften

Die Klasse *ToolPanel* verfügt u.a. über diese Eigenschaften:

Eigenschaft	Datentyp	Beschreibung
Animated	Boolean	Legt fest, ob die Symbolleisten animiert sind.
Border	Boolean	Ermittelt oder setzt, ob das ToolPanel einen Rahmen hat.
Count	Integer	Liefert oder setzt die Anzahl der Symbolleisten.
Font	Font	Gibt die Schriftart zurück oder legt sie fest, die zum Zeichnen von Text im Steuerelement verwendet wird.
Index	Integer	Liefert oder setzt die aktuelle Symbolleiste.
Picture	Picture	Gibt das in der Schaltfläche der aktuellen Symbolleiste angezeigte Symbol (Datentyp: Picture) zurück oder legt es fest.
Text	String	Liefert oder setzt die Beschriftung der aktuellen Symbolleiste.

Tabelle 18.4.1.1 : Eigenschaften der Klasse ToolPanel

18.4.2 Ereignisse

Die Klasse *ToolPanel* verfügt nur über zwei relevante Ereignisse:

Ereignis	Beschreibung
DbfClick ()	Dieses Ereignis wird ausgelöst, wenn der Benutzer schnell zweimal auf das Steuerelement klickt.
Click ()	Dieses Ereignis wird ausgelöst, wenn der Benutzer eine Symbolleiste (Toolbar) im ToolPanel auswählt.

Tabelle 18.4.2.1 : Ereignisse der Klasse ToolPanel

18.4.3 Projekt

Vorgestellt wird Ihnen ein kleines Projekt, bei dem Sie in einem ToolPanel eine von drei Toolbars auswählen können, die unterschiedliche Funktionalitäten bereitstellen:




Abbildung 18.4.3.1: Auswahl: Farbe




Abbildung 18.4.3.2: Auswahl: Datum




Abbildung 18.4.3.3: Auswahl: Datei

Für das ToolPanel können Sie einerseits dessen grundlegende Eigenschaften in der IDE festlegen und andererseits nach der Festlegung, wie viele Toolbars das ToolPanel aufnehmen soll, auch die wesentlichen Eigenschaften der einzelnen Toolbars.

Der Quelltext sieht sehr übersichtlich aus, weil nur noch die wesentliche Funktionalität der einzelnen Toolbars festzulegen ist (Auswahl Farbe, Auswahl Datum1 und Datum2 sowie Auswahl Datei):

```

' Gambas class file
Public Sub Form_Open()

    FMain.Resizable = False
    ColorChooser1.Value = &HCF3FFF

    FileChooser1.Root = User.Home
    FileChooser1.Multi = False
    FileChooser1.Filter = ["*.png;*.jpg;*.gif", "Picture files", "*.txt;*.xml;*.conf", "Text files"]
    FileChooser1.FilterIndex = 0      '-- Display images with the extensions png, jp* and gif first
    '-- FileChooser1.FilterIndex = -1  '-- Display all files
    '-- FileChooser1.ShowFile = True  '-- Default-Value is 'True'
    '-- FileChooser1.ShowDirectory = False '-- Default-Value is 'False'
    '-- FileChooser1.ShowButton = False '-- Default-Value is 'False'
    FileChooser1.ShowPreview = True  '-- Default-Value is 'False'
End

Public Sub DateChooser1_Change()

    lblDifferenceInDays.Text = RunDateDiff(DateChooser1.Value, DateChooser2.Value)
End

Public Sub DateChooser2_Change()

    lblDifferenceInDays.Text = RunDateDiff(DateChooser1.Value, DateChooser2.Value)
End

Private Sub RunDateDiff(hDate1 As Date, hDate2 As Date) As Integer

    Return Abs(DateDiff(hDate1, hDate2, gb.Day)) + 1
    
```

```
End
Public Sub ColorChooser1_Change()
    lblColor.Text = "&H" & Hex(ColorChooser1.Value)
End
Public Sub FileChooser1_Change()
    lblFilename.Text = ""
    If FileChooser1.SelectedPath Then lblFilename.Text = FileChooser1.SelectedPath
End
Public Sub FileChooser1_Activate()
    If FileChooser1.Multi = False Then
        lblFilename.Text = FileChooser1.SelectedPath
        '-- Message.Info("Selected file:<hr>" & FileChooser1.SelectedPath)
        '-- Desktop.Open(FileChooser1.DirView.Current &/ FileChooser1.FileView.Selection[0])
        Desktop.Open(FileChooser1.SelectedPath)
    Endif
End
```

Eine Besonderheit besteht bei der Auswahl einer Datei darin, dass nach einem Doppel-Klick auf eine Datei die ausgewählte Datei mit dem Standard-Editor geöffnet und deren Inhalt angezeigt wird.

Ihnen wird der komplette Quelltext in einem Projekt-Archiv im Download-Bereich zur Verfügung gestellt.